

# NAG Library Routine Document

## F04CHF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F04CHF computes the solution to a complex system of linear equations  $AX = B$ , where  $A$  is an  $n$  by  $n$  Hermitian matrix and  $X$  and  $B$  are  $n$  by  $r$  matrices. An estimate of the condition number of  $A$  and an error bound for the computed solution are also returned.

### 2 Specification

SUBROUTINE F04CHF (UPLO, N, NRHS, A, LDA, IPIV, B, LDB, RCOND, ERBND, &  
IFAIL)

INTEGER N, NRHS, LDA, IPIV(N), LDB, IFAIL  
REAL (KIND=nag\_wp) RCOND, ERBND  
COMPLEX (KIND=nag\_wp) A(LDA,\*), B(LDB,\*)  
CHARACTER(1) UPLO

### 3 Description

The diagonal pivoting method is used to factor  $A$  as  $A = UDU^H$ , if UPLO = 'U', or  $A = LDL^H$ , if UPLO = 'L', where  $U$  (or  $L$ ) is a product of permutation and unit upper (lower) triangular matrices, and  $D$  is Hermitian and block diagonal with 1 by 1 and 2 by 2 diagonal blocks. The factored form of  $A$  is then used to solve the system of equations  $AX = B$ .

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

### 5 Arguments

- 1: UPLO – CHARACTER(1) *Input*  
*On entry:* if UPLO = 'U', the upper triangle of the matrix  $A$  is stored.  
 If UPLO = 'L', the lower triangle of the matrix  $A$  is stored.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*  
*On entry:* the number of linear equations  $n$ , i.e., the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: NRHS – INTEGER *Input*  
*On entry:* the number of right-hand sides  $r$ , i.e., the number of columns of the matrix  $B$ .  
*Constraint:*  $NRHS \geq 0$ .

- 4: A(LDA,\*) – COMPLEX (KIND=nag\_wp) array Input/Output  
**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .  
*On entry:* the  $n$  by  $n$  Hermitian matrix  $A$ .  
 If UPLO = 'U', the leading  $N$  by  $N$  upper triangular part of the array A contains the upper triangular part of the matrix  $A$ , and the strictly lower triangular part of A is not referenced.  
 If UPLO = 'L', the leading  $N$  by  $N$  lower triangular part of the array A contains the lower triangular part of the matrix  $A$ , and the strictly upper triangular part of A is not referenced.  
*On exit:* if IFAIL  $\geq 0$ , the block diagonal matrix  $D$  and the multipliers used to obtain the factor  $U$  or  $L$  from the factorization  $A = UDU^H$  or  $A = LDL^H$  as computed by F07MRF (ZHETRF).
- 5: LDA – INTEGER Input  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F04CHF is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 6: IPIV(N) – INTEGER array Output  
*On exit:* if IFAIL  $\geq 0$ , details of the interchanges and the block structure of  $D$ , as determined by F07MRF (ZHETRF).  
 If  $IPIV(k) > 0$ , then rows and columns  $k$  and  $IPIV(k)$  were interchanged, and  $d_{kk}$  is a 1 by 1 diagonal block;  
 if UPLO = 'U' and  $IPIV(k) = IPIV(k-1) < 0$ , then rows and columns  $k-1$  and  $-IPIV(k)$  were interchanged and  $d_{k-1:k, k-1:k}$  is a 2 by 2 diagonal block;  
 if UPLO = 'L' and  $IPIV(k) = IPIV(k+1) < 0$ , then rows and columns  $k+1$  and  $-IPIV(k)$  were interchanged and  $d_{k:k+1, k:k+1}$  is a 2 by 2 diagonal block.
- 7: B(LDB,\*) – COMPLEX (KIND=nag\_wp) array Input/Output  
**Note:** the second dimension of the array B must be at least  $\max(1, NRHS)$ .  
*On entry:* the  $n$  by  $r$  matrix of right-hand sides  $B$ .  
*On exit:* if IFAIL = 0 or  $N + 1$ , the  $n$  by  $r$  solution matrix  $X$ .
- 8: LDB – INTEGER Input  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F04CHF is called.  
*Constraint:*  $LDB \geq \max(1, N)$ .
- 9: RCOND – REAL (KIND=nag\_wp) Output  
*On exit:* if no constraints are violated, an estimate of the reciprocal of the condition number of the matrix  $A$ , computed as  $RCOND = 1 / \left( \|A\|_1 \|A^{-1}\|_1 \right)$ .
- 10: ERRBND – REAL (KIND=nag\_wp) Output  
*On exit:* if IFAIL = 0 or  $N + 1$ , an estimate of the forward error bound for a computed solution  $\hat{x}$ , such that  $\|\hat{x} - x\|_1 / \|x\|_1 \leq ERRBND$ , where  $\hat{x}$  is a column of the computed solution returned in the array B and  $x$  is the corresponding column of the exact solution  $X$ . If RCOND is less than **machine precision**, then ERRBND is returned as unity.
- 11: IFAIL – INTEGER Input/Output  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or  $1$  is recommended. If the output of error messages is undesirable, then the value  $1$  is recommended. Otherwise, if you are not familiar with this argument, the recommended value is  $0$ . **When the value  $-1$  or  $1$  is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL =  $0$  unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL =  $0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL >  $0$  and IFAIL  $\leq$  N

Diagonal block  $\langle value \rangle$  of the block diagonal matrix is zero. The factorization has been completed, but the solution could not be computed.

IFAIL = N + 1

A solution has been computed, but RCOND is less than *machine precision* so that the matrix  $A$  is numerically singular.

IFAIL =  $-1$

On entry, UPLO not one of 'U' or 'u' or 'L' or 'l': UPLO =  $\langle value \rangle$ .

IFAIL =  $-2$

On entry, N =  $\langle value \rangle$ .  
Constraint: N  $\geq$  0.

IFAIL =  $-3$

On entry, NRHS =  $\langle value \rangle$ .  
Constraint: NRHS  $\geq$  0.

IFAIL =  $-5$

On entry, LDA =  $\langle value \rangle$  and N =  $\langle value \rangle$ .  
Constraint: LDA  $\geq$  max(1, N).

IFAIL =  $-8$

On entry, LDB =  $\langle value \rangle$  and N =  $\langle value \rangle$ .  
Constraint: LDB  $\geq$  max(1, N).

IFAIL =  $-99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL =  $-399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL =  $-999$

Dynamic memory allocation failed.

The real allocatable memory required is  $N$ , and the complex allocatable memory required is  $\max(2 \times N, LWORK)$ , where  $LWORK$  is the optimum workspace required by  $F07MNF$  ( $ZHESV$ ). If this failure occurs it may be possible to solve the equations by calling the packed storage version of  $F04CHF$ ,  $F04CJF$ , or by calling  $F07MNF$  ( $ZHESV$ ) directly with less than the optimum workspace (see Chapter F07).

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

The computed solution for a single right-hand side,  $\hat{x}$ , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and  $\epsilon$  is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where  $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$ , the condition number of  $A$  with respect to the solution of the linear equations.  $F04CHF$  uses the approximation  $\|E\|_1 = \epsilon \|A\|_1$  to estimate  $ERRBND$ . See Section 4.4 of Anderson *et al.* (1999) for further details.

## 8 Parallelism and Performance

$F04CHF$  makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations required to solve the equations  $AX = B$  is proportional to  $(\frac{1}{3}n^3 + 2n^2r)$ . The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization.

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

Routine  $F04DHF$  is for complex symmetric matrices, and the real analogue of  $F04CHF$  is  $F04BHF$ .

## 10 Example

This example solves the equations

$$AX = B,$$

where  $A$  is the Hermitian indefinite matrix

$$A = \begin{pmatrix} -1.84 & 0.11 - 0.11i & -1.78 - 1.18i & 3.91 - 1.50i \\ 0.11 + 0.11i & -4.63 & -1.84 + 0.03i & 2.21 + 0.21i \\ -1.78 + 1.18i & -1.84 - 0.03i & -8.87 & 1.58 - 0.90i \\ 3.91 + 1.50i & 2.21 - 0.21i & 1.58 + 0.90i & -1.36 \end{pmatrix}$$

and

$$B = \begin{pmatrix} 2.98 - 10.18i & 28.68 - 39.89i \\ -9.58 + 3.88i & -24.79 - 8.40i \\ -0.77 - 16.05i & 4.23 - 70.02i \\ 7.79 + 5.48i & -35.39 + 18.01i \end{pmatrix}.$$

An estimate of the condition number of  $A$  and an approximate error bound for the computed solutions are also printed.

### 10.1 Program Text

```

Program f04chfe

!      F04CHF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: f04chf, nag_wp, x04dbf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: errbnd, rcond
Integer                     :: i, ierr, ifail, lda, ldb, n, nrhs
!      .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: a(:,,:), b(:,,:)
Integer, Allocatable        :: ipiv(:)
Character (1)               :: clabs(1), rlabs(1)
!      .. Executable Statements ..
Write (nout,*) 'F04CHF Example Program Results'
Write (nout,*)
Flush (nout)
!      Skip heading in data file
Read (nin,*)
Read (nin,*) n, nrhs
lda = n
ldb = n
Allocate (a(lda,n),b(ldb,nrhs),ipiv(n))
!      Read the upper triangular part of A from data file
Read (nin,*)(a(i,i:n),i=1,n)

!      Read B from data file
Read (nin,*)(b(i,1:nrhs),i=1,n)

!      Solve the equations AX = B for X

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 1
Call f04chf('Upper',n,nrhs,a,lda,ipiv,b,ldb,rcond,errbnd,ifail)

If (ifail==0) Then
!      Print solution, estimate of condition number and approximate
!      error bound

ierr = 0
Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed',' ','Solution',
           'Integer',rlabs,'Integer',clabs,80,0,ierr)

Write (nout,*)
Write (nout,*) 'Estimate of condition number'
Write (nout,99999) 1.0E0_nag_wp/rcond
Write (nout,*)
Write (nout,*) 'Estimate of error bound for computed solutions'
Write (nout,99999) errbnd
Else If (ifail==n+1) Then
!      Matrix A is numerically singular. Print estimate of

```

```

!      reciprocal of condition number and solution
      Write (nout,*)
      Write (nout,*) 'Estimate of reciprocal of condition number'
      Write (nout,99999) rcond
      Write (nout,*)
      Flush (nout)

      ierr = 0
      Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed',' ','Solution',      &
        'Integer',rlabs,'Integer',clabs,80,0,ierr)

      Else If (ifail>0 .And. ifail<=n) Then
!      The upper triangular matrix U is exactly singular.  Print
!      details of factorization
      Write (nout,*)
      Flush (nout)

      ierr = 0
      Call x04dbf('Upper','Non-unit diagonal',n,n,a,lda,'Bracketed',' ',      &
        'Details of factorization','Integer',rlabs,'Integer',clabs,80,0,      &
        ierr)

!      Print pivot indices
      Write (nout,*)
      Write (nout,*) 'Pivot indices'
      Write (nout,99998) ipiv(1:n)
      Else
        Write (nout,99997) ifail
      End If

99999 Format (8X,1P,E9.1)
99998 Format ((1X,7I11))
99997 Format (1X,' ** F04CHF returned with IFAIL = ',I5)
      End Program f04chfe

```

## 10.2 Program Data

F04CHF Example Program Data

```

      4              2              : n, nrhs

( -1.84,  0.00) (  0.11, -0.11) ( -1.78, -1.18) (  3.91, -1.50)
              ( -4.63,  0.00) ( -1.84,  0.03) (  2.21,  0.21)
              ( -8.87,  0.00) (  1.58, -0.90)
              ( -1.36,  0.00) : matrix A

(  2.98,-10.18) ( 28.68,-39.89)
( -9.58,  3.88) (-24.79, -8.40)
( -0.77,-16.05) (  4.23,-70.02)
(  7.79,  5.48) (-35.39, 18.01)      : matrix B

```

## 10.3 Program Results

F04CHF Example Program Results

```

Solution
              1              2
1 (  2.0000,  1.0000) ( -8.0000,  6.0000)
2 (  3.0000, -2.0000) (  7.0000, -2.0000)
3 ( -1.0000,  2.0000) ( -1.0000,  5.0000)
4 (  1.0000, -1.0000) (  3.0000, -4.0000)

Estimate of condition number
        6.7E+00

Estimate of error bound for computed solutions
        7.4E-16

```